

Program Transformations in the Delay Monad

A Case Study for Coinduction via Copatterns and Sized Types



Edoardo Marangoni
University of Milan

A thesis submitted for the degree of
Master of Science

`datetime(year: 2023, month: 9, day: 18)`

*“...I can hardly understand, for instance, how a young man can decide to ride over
to the next village without being afraid that, quite apart from accidents,
even the span of a normal life that passes happily may be totally
insufficient for such a ride.”*

Franz Kafka

NO GENERATIVE ARTIFICIAL INTELLIGENCE WAS USED IN
THIS WORK.

Content

Chapter 1 Introduction

Chapter 2 Induction and coinduction

2.1 Infinite datatypes	3
2.2 Infinite proofs	3
2.3 Relation with fixed points	3

Chapter 3 Agda

3.1 Dependent types	5
3.2 Termination and productivity	5
3.3 Sized types	5

Chapter 4 The partiality monad

4.1 Monads	7
4.1.1 Formal definition	8
4.2 The Delay monad	8
4.3 Bisimilarity	9

Chapter 5 The Imp programming language

5.1 Introduction	13
5.1.1 Syntax	13
5.1.2 Properties of stores	14
5.1.3 Properties of expressions	16
5.2 Semantics	17
5.2.1 Arithmetic expressions	18
5.2.2 Boolean expressions	19
5.2.3 Commands	19
5.2.4 Properties of the interpreter	19
5.3 Analyses and optimizations	20
5.3.1 Definite initialization analysis	20
5.3.2 Pure constant folding optimization	25

Appendix A Proofs

A.1 The partiality monad	29
A.1.1 Bisimilarity	29
A.2 The Imp programming language	30
A.2.1 Properties of stores	30

A.2.2 Semantics	30
-----------------------	----

Bibliography

CHAPTER I

Introduction

CHAPTER 2

Induction and coinduction

2.1 Infinite datatypes

2.2 Infinite proofs

2.3 Relation with fixed points

In this chapter we will introduce the Agda programming language.

1. the shortest history of proof assistants ever
2. what makes agda useful, i.e., dependent types

3.1 Dependent types

3.2 Termination and productivity

3.3 Sized types

The partiality monad

In this chapter we introduce the concept of monad and then describe a particular kind of monad, the *partiality monad*, which will be used throughout the work.

4.1 Monads

In 1989, computer scientist Eugenio Moggi published a paper (Moggi 1989) in which the term *monad*, which was already used in the context of mathematics and, in particular, category theory, was given meaning in the context of functional programming. Explaining monads is, arguably, one of the most discussed topics in the pedagogy of computer science, and tons of articles, blog posts and books try to explain the concept of monad in various ways.

A monad is a datatype equipped with (at least) two functions, `bind` (often `_>=>_`) and `unit`; in general, we can see monads as a structure used to combine computations. One of the most trivial instance of monad is the Maybe monad, which we now present to investigate what monads are: in Agda, the Maybe monad is composed of a datatype

```
data Maybe {a} (A : Set a) : Set a where
  just  : A → Maybe A
  nothing : Maybe A
```

and two functions representing its monadic features:

```
unit : A → Maybe A
unit = just

_>=>_ : Maybe A → (A → Maybe B) → Maybe B
nothing >=> f = nothing
just a   >=> f = f a
```

The Maybe monad is a structure that represents how to deal with computations that may result in a value but may also result in nothing; in general, the line of reasoning for monads is exactly this, they are a means to model a behaviour of the execution, or **effects**: in fact, they're also called “computation builders” in the context of programming. Let's give an example:

```
h : Maybe N → Maybe N
h x = x >=> λ v → just (v + 1)
```

Without `bind`, `h` would be a match on the value of `x`: if `x` is just `v` then do something, otherwise, if `x` is nothing, return nothing. The underlying idea of monads in the context of computer science, as explained by Moggi in (Moggi 1989), is to describe “notions of computations” that may have consequences comparable to *side effects* of imperative programming languages in pure functional languages.

4.1.1 Formal definition

We will now give a formal definition of what monads are. They’re usually understood in the context of category theory and in particular *Kleisli triples*; here, we give a minimal definition inspired by (Kohl and Schwaiger 2021).

Definition 4.1.1.1 (Monad): Let A, B and C be types. A monad M is defined as the triple $(\mathbb{m}, \text{unit}, _ \gg= _)$ where \mathbb{m} is a monadic constructor denoting some side-effect or impure behaviour; $\text{unit} : A \rightarrow M A$ represents the identity function and $_ \gg= _ : M A \rightarrow (A \rightarrow M B) \rightarrow M B$ is used for monadic composition.

The triple must satisfy the following laws.

1. **(left identity)** For every $x : A$ and $f : A \rightarrow M B$, $\text{unit } x \gg= f \equiv f x$;
2. **(right identity)** For every $mx : M A$, $mx \gg= \text{unit} \equiv mx$; and
3. **(associativity)** For every $mx : M A$, $f : A \rightarrow M B$ and $g : B \rightarrow M C$,
 $(mx \gg= f) \gg= g \equiv mx \gg= (\lambda my \rightarrow f my \gg= g)$

4.2 The Delay monad

In 2005, computer scientist Venanzio Capretta introduced the `Delay` monad to represent recursive (thus potentially infinite) computations in a coinductive (and monadic) fashion (Capretta 2005). As described in (Abel and Chapman 2014), the `Delay` type is used to represent computations whose result may be available with some *delay* or never be returned at all: the `Delay` type has two constructors; one, `now`, contains the result of the computation. The second, `later`, embodies one “step” of delay and, of course, an infinite (coinductive) sequence of `later` indicates a non-terminating computation, practically making non-termination (partiality) an effect, taking the perspective of

In Agda, the `Delay` type is defined as follows (using *sizes*, see Chapter 3.3):

```
data Delay {ℓ} (A : Set ℓ) (i : Size) : Set ℓ where
  now   : A → Delay A i
  later : Thunk (Delay A) i → Delay A i
```

We equip with the following `bind` function:


```

bind : ∀ {i} → Delay A i → (A → Delay B i) → Delay B i
bind (now a) f = f a
bind (later d) f = later λ where .force → bind (d .force) f

```

In words, what `bind` does, is this: given a `Delay A i x`, it checks whether `x` contains an immediate result (i.e., `x ≡ now a`) and, if so, it applies the function `f`; if, otherwise, `x` is a step of delay, (i.e., `x ≡ later d`), `bind` delays the computation by wrapping the observation of `d` (represented as `d .force`) in the `later` constructor. Of course, this is the only possible definition: for example, `bind' (later d) f = bind' (d .force) f` would not pass the termination and productivity checker; in fact, take the `never` term as shown in Listing 1: of course, `bind' never f` would never terminate.

```

never : ∀ {i} → Delay A i
never = later λ where .force → never

```

Listing 1: Non-terminating term in the Delay monad

We might however argue that `bind` as well `never` terminates, in fact `never` *never yields a value* by definition; this is correct, but the two views on non-termination are radically different. The detail is that `bind'` observes the whole of `never` immediately, while `bind` leaves to the observer the job of actually inspecting what the result of `bind x f` is, and this is the utility of the `Delay` datatype and its monadic features.

4.3 Bisimilarity

An important notion relating terms of `Delay` type is that of *bisimilarity*. **Strong** bisimilarity relates diverging computations and computations that converge to the same value using the same number of steps (Danielsson 2012); the formal definition we give is from (Chapman, Uustalu, and Veltri 2015), and is shown in Definition 4.3.1 properties of this relation are given in Theorem 4.3.1.

A less strict relation is **weak** bisimilarity: it equally relates diverging terms (coinductively), but it also allows the relation between computations that converge to the same value but in different number of steps.

Definition 4.3.1 (Strong bisimilarity): Let \mathcal{A} be a type for which there exists an equivalence relation $R_{\mathcal{A}}$, and let a_1 and a_2 be two terms of type \mathcal{A} . Furthermore, let x_1 and x_2 be two delayed terms. Then, we define the strong bisimilarity relation $\sim_{\mathcal{A}}$ as

$$\frac{a_1 R a_2}{\approx\text{-now: now } a_1 \sim_A \text{ now } a_2} \quad \frac{x_1 \sim_A x_2}{\approx\text{-later: later } x_1 \sim_A \text{ later } x_2}$$

In Agda:

```
data Bisim {a b r} {A : Set a} {B : Set b} (R : A → B → Set r) i :
  (xs : Delay A ∞) (ys : Delay B ∞) → Set (a ∪ b ∪ r) where
  now   : ∀ {x y} → R x y → Bisim R i (now x) (now y)
  later : ∀ {xs ys} → Thunk^R (Bisim R) i xs ys → Bisim R i (later xs) (later ys)
```

Theorem 4.3.1 (Strong bisimilarity is an equivalence relation): For every equivalence relation \mathcal{A} , the strong bisimilarity relation $\sim_{\mathcal{A}}$ is an equivalence relation. Furthermore, strong bisimilarity is a transitive relation. In Agda:

```
reflexive : Reflexive R → ∀ {i} → Reflexive (Bisim R i)
symmetric : Sym P Q → ∀ {i} → Sym (Bisim P i) (Bisim Q i)
transitive : Trans P Q R → ∀ {i} → Trans (Bisim P i) (Bisim Q i) (Bisim R i)
```

Definition 4.3.2 (Weak bisimilarity): Let \mathcal{A} be a type for which there exists an equivalence relation $R_{\mathcal{A}}$, and let a_1 and a_2 be two terms of type \mathcal{A} . Furthermore, let x_1 and x_1 be two delayed terms. Then, we define the weak bisimilarity relation $\sim_{\mathcal{A}}$ as

$$\frac{a_1 R a_2}{\approx\text{-now: now } a_1 \sim_{\mathcal{A}} \text{ now } a_2} \quad \frac{\text{force } x_1 \sim_{\mathcal{A}} \text{ force } x_2}{\approx\text{-later: later } x_1 \sim_{\mathcal{A}} \text{ later } x_2}$$

$$\frac{\text{force } x_1 \sim_a x_2}{\approx\text{-later-l: later } x_1 \sim_a x_2} \quad \frac{x_1 \sim_a \text{ force } x_2}{\approx\text{-later-r: } x_1 \sim_a \text{ later } x_2}$$

In Agda:

```
data Bisim {a b r} {A : Set a} {B : Set b} (R : A → B → Set r) i :
  (xs : Delay A ∞) (ys : Delay B ∞) → Set (a ∪ b ∪ r) where
  now   : ∀ {x y} → R x y → Bisim R i x y
  later : ∀ {xs ys} → Thunk^R (Bisim R) i xs ys → Bisim R i (later xs) (later ys)
  later_l : ∀ {xs ys} → Bisim R i (force xs) ys → Bisim R i (later xs) ys
  later_r : ∀ {xs ys} → Bisim R i xs (force ys) → Bisim R i xs (later ys)
```

Theorem 4.3.2 (Weak bisimilarity is an equivalence relation): For every equivalence relation \mathcal{A} , the weak bisimilarity relation $\sim_{\mathcal{A}}$ is an equivalence relation. In Agda:

```
reflexive : Reflexive R → ∀ {i} → Reflexive (WeakBisim R i)
symmetric : Sym P Q → ∀ {i} → Sym (Bisim P i) (WeakBisim Q i)
```

It's also trivial to show that strong bisimilarity implies weak bisimilarity. We can also now prove monad laws up to strong bisimilarity, as shown in Theorem 4.3.3.

Theorem 4.3.3 (Delay is a monad): The triple (Delay, now, bind) is a monad and respects monad laws up to bisimilarity. In Agda:

```
left-identity : ∀ {i} (x : A) (f : A → Delay B i) → bind (now x) f ≡ f x
right-identity : ∀ {i} (x : Delay A ∞) → i ⊢ x ≍ now ≈ x
associativity : ∀ {i} {x : Delay A ∞} {f : A → Delay B ∞} {g : B → Delay C ∞}
  → i ⊢ (x ≍ f) ≍ g ≈ x ≍ λ y → (f y ≍ g)
```


The Imp programming language

In this chapter we will go over the implementation of a simple imperative language called **Imp**, as described in (Pierce et al. 2023). After defining its syntax, we will give rules for its semantics and show its implementation in Agda. After this introductory work, we will discuss analysis and optimization of Imp programs.

5.1 Introduction

The Imp language was devised to work as a simple example of an imperative language; albeit having a handful of syntactic constructs, it's clearly a Turing complete language.

5.1.1 Syntax

The syntax of the Imp language is can be described in a handful of EBNF rules, as shown in Table 1.

$$\begin{aligned}
 \mathbf{aexp} &:= n \mid \text{id} \mid a_1 + a_2 \\
 \mathbf{bexp} &:= b \mid a_1 < a_2 \mid \neg b \mid b_1 \wedge b_2 \\
 \mathbf{command} &:= \text{skip} \mid \text{id} \leftarrow \mathbf{aexp} \mid c_1; c_2 \mid \text{if } \mathbf{bexp} \text{ then } c_1 \text{ else } c_2 \mid \text{while } \mathbf{bexp} \text{ do } c
 \end{aligned}$$

Table 1: Syntax rules for the Imp language

The syntactic elements of this language are three: *commands*, *arithmetic expressions* and *boolean expressions*. Given its simple nature, it's easy to give an abstract representation for its concrete syntax: all three can be represented with simple datatypes enclosing all the information of the syntactic rule.

Another important atomic element of Imp are *identifiers*. Identifiers can mutate in time and, when misused, cause errors during the execution of programs: in fact, there is no way to enforce the programmer to use only initialized identifiers merely by syntax rules – it would take a context-sensitive grammar to achieve so, at least. A concept related to identifiers is that of *stores*, which are conceptually instantaneous descriptions of the state of identifiers in the ideal machine executing the program.

We now show how we implemented the syntactic elements of Imp in Agda and show a handful of trivial properties: in Listing 2 we show the datatypes for identifiers and stores, while in Listing 3 we show the datatypes for the other syntactic constructs.

```

Ident : Set      Store : Set
Ident = String  Store = Ident → Maybe ℤ

```

Listing 2: Datatypes for identifiers and stores

Notice that the implementation of Stores reflect the behaviour described earlier in that they are intended as functions from Ident to Maybe \mathbb{Z} .

```

data AExp : Set where
  const : (n : ℤ) → AExp
  var    : (id : Ident) → AExp
  plus   : (a1 a2 : AExp) → AExp

data BExp : Set where
  const : (b : Bool) → BExp
  le     : (a1 a2 : AExp) → BExp
  not    : (b : BExp) → BExp
  and    : (b1 b2 : BExp) → BExp

data Command : Set where
  skip    : Command
  assign  : (id : Ident) → (a : AExp) → Command
  seq     : (c1 c2 : Command) → Command
  ifelse  : (b : BExp) → (c1 c2 : Command) → Command
  while   : (b : BExp) → (c : Command) → Command

```

Listing 3: Datatype for expressions of Imp

5.1.2 Properties of stores

The first properties we show regard stores. We equip stores with the trivial operations of adding an identifier, merging two stores and joining two stores, as shown in Listing 2.

```

empty : Store
empty = λ _ → nothing
update : (id1 : Ident) → (v : ℤ) → (s : Store) → Store
update id1 v s id2
  with id1 == id2
... | true = (just v)
... | false = (s id2)
join : (s1 s2 : Store) → Store
join s1 s2 id
  with (s1 id)
... | just v = just v
... | nothing = s2 id
merge : (s1 s2 : Store) → Store
merge s1 s2 =
  λ id → (s1 id) >=
  λ v1 → (s2 id) >=
  λ v2 → if (⊥ v1 ≐ v2 ⊥) then just v1 else nothing

```

Listing 2: Operations on stores

A trivial property of stores is that of unvalued inclusion, that is, a property stating that if an identifier has a value in a store σ_1 , then it also has a value (not necessarily the same) in another store σ_2 :

Property 5.1.2.1 (Unvalued store inclusion): Let σ_1 and σ_2 be two stores. We define the unvalued inclusion between them as

$$\forall \text{id}, (\exists z, \sigma_1 \text{id} \equiv \text{just } z) \rightarrow (\exists z, \sigma_2 \text{id} \equiv \text{just } z) \quad (1)$$

and we denote it with $\sigma_1 \stackrel{u}{\subseteq} \sigma_2$. In Agda:

We equip Property 5.1.2.1 with a notion of transitivity.

Theorem 5.1.2.1 (Transitivity of unvalued store inclusion): Let σ_1, σ_2 and σ_3 be three stores. Then

$$\sigma_1 \stackrel{u}{\subseteq} \sigma_2 \wedge \sigma_2 \stackrel{u}{\subseteq} \sigma_3 \rightarrow \sigma_1 \stackrel{u}{\subseteq} \sigma_3 \quad (2)$$

In Agda:

The operations we define on stores are multiple: adding an identifier paired with a value to a store, removing an identifier from a store, joining stores and merging stores. We now define notations:

1. **in-store predicate** let $\text{id} : \text{Ident}$ and $\sigma : \text{Store}$. To say that id is in σ we write $\text{id} \in \sigma$; in other terms, it's the same as $\exists v \in \mathbb{Z}, \sigma \text{id} \equiv \text{just } v$.
2. **empty store** we define the empty store as \emptyset . For this special store, it is always $\forall \text{id}, \text{id} \in \emptyset \rightarrow \perp$ or $\forall \text{id}, \emptyset \text{id} \equiv \text{nothing}$.
3. **adding an identifier** let $\text{id} : \text{Ident}$ be an identifier and $v : \mathbb{Z}$ be a value. We denote the insertion of the pair (id, v) in a store σ as $(\text{id}, v) \mapsto \sigma$.
4. **joining two stores** let σ_1 and σ_2 be two stores. We define the store that contains an id if $\text{id} \in \sigma_1$ or $\text{id} \in \sigma_2$ as $\sigma_1 \cup \sigma_2$. Notice that the join operation is not commutative, as it may be that

$$\exists \text{id}, \exists v_1, \exists v_2, v_1 \neq v_2 \wedge \sigma_1 \text{id} \equiv \text{just } v_1 \wedge \sigma_2 \text{id} \equiv \text{just } v_2 \quad (3)$$

5. **merging two stores** let σ_1 and σ_2 be two stores. We define the store that contains an id if and only if $\sigma_1 \text{id} \equiv \text{just } v$ and $\sigma_2 \text{id} \equiv \text{just } v$ as $\sigma_1 \cap \sigma_2$.

5.1.3 Properties of expressions

The properties of expressions we show here regard the syntactic relation between elements. The property we define is that of *subterm relation*. In Agda, as will be shown in the definitions, these properties are implemented as datatypes. Properties 5.1.3.3, 5.1.3.2 and 5.1.3.1 will be used later to relate semantic aspects of subterms with that of the containing term itself or vice versa.

Property 5.1.3.1 (Arithmetic subterms): Let a_1 and a_2 be arithmetic expressions.

Then

$$a_1 \stackrel{a}{\sqsubseteq} \text{plus } a_1 \ a_2 \quad a_2 \stackrel{a}{\sqsubseteq} \text{plus } a_1 \ a_2$$

In Agda:

Property 5.1.3.2 (Boolean subterms): Let a_1 and a_2 be arithmetic expressions and b_1 and b_2 be boolean expressions.

Then

$$\begin{aligned} a_1 \stackrel{b}{\sqsubseteq} \text{le } a_1 \ a_2 & \quad a_2 \stackrel{b}{\sqsubseteq} \text{le } a_1 \ a_2 \\ b_1 \stackrel{b}{\sqsubseteq} \text{and } b_1 \ b_2 & \quad b_2 \stackrel{b}{\sqsubseteq} \text{and } b_1 \ b_2 \\ b_1 \stackrel{b}{\sqsubseteq} \text{not } b_1 & \end{aligned}$$

In Agda:

`data`

```
_⊆b_ : {A : Set} → A → BExp → Set where
  not : (b : BExp) → b ⊆b (not b)
  and-l : (b1 b2 : BExp) → b1 ⊆b (and b1 b2)
  and-r : (b1 b2 : BExp) → b2 ⊆b (and b1 b2)
  le-l : (a1 a2 : AExp) → a1 ⊆b (le a1 a2)
  le-r : (a1 a2 : AExp) → a2 ⊆b (le a1 a2)
```

In Agda:

Property 5.1.3.3 (Command subterms): Let id be an identifier, a be an arithmetic expressions, b be a boolean expression and c_1 and c_2 be commands.

Then

$$\begin{array}{llll} a \stackrel{c}{\in} \text{assign } \text{id } a & c_1 \stackrel{c}{\in} \text{seq } c_1 c_2 & c_2 \stackrel{c}{\in} \text{seq } c_1 c_2 & b \stackrel{c}{\in} \text{if } b c_1 c_2 \\ c_1 \stackrel{c}{\in} \text{if } b c_1 c_2 & c_2 \stackrel{c}{\in} \text{if } b c_1 c_2 & b \stackrel{c}{\in} \text{while } b c_1 & c_1 \stackrel{c}{\in} \text{while } b c_1 \end{array}$$

In Agda:

5.2 Semantics

Having understood the concrete and abstract syntax of Imp, we can move to the meaning of Imp programs. We'll explore the operational semantics of the language using the formalism of inference rules, then we'll show the implementation of the semantics (as an interpreter) for these rules.

Before describing the rules of the semantics, we will give a brief explanation of what we expect to be the result of the evaluation of an Imp program. As shown in Table 1, Imp programs are composed of three entities: arithmetic expression, boolean expression and commands.

```
true then skip else 1
```

Listing 3: A simple Imp
program

An example of Imp program is shown in Listing 3: note that is technically not well-typed, but we don't care about this now. In general, we can expect the evaluation of an Imp program to terminate in some kind value or diverge, but it might also **fail**: this is the case when an uninitialized variable is used, as we mentioned in Chapter 5.1.1.

We could model other kinds of failures, both deriving from static analysis (such as failures of type-checking) or from the dynamic execution of the program, but we chose to model this kind of behaviour only: an example of this can be seen in Listing 4.

```
while true do x := y
```

Listing 4: A failing (not diverging!) Imp program

We can now introduce the formal notation we will use to describe the semantics of Imp programs. We already introduced the concept of store, which keeps track of the

mutation of identifiers and their value during the execution of the program. We write $c, \sigma \Downarrow \sigma_1$ to mean that the program c , when evaluated starting from the context σ , converges to the store σ_1 .

We write $c, \sigma \not\Downarrow$ to say that the program c , when evaluated in context σ , does not converge to a result but, instead, execution gets stuck (that is, an unknown identifier is used).

The last possibility is for the execution to diverge, $c, \sigma \Uparrow$: this means that the evaluation of the program never stops and while no state of failure is reached no result is ever produced. An example of this behaviour is seen when evaluating Listing 5.

while true do skip

Listing 5: A diverging Imp program

We're now able to give inference rules for each construct of the Imp language: we'll start from simple ones, that is arithmetic and boolean expressions, and we'll then move to commands.

5.2.1 Arithmetic expressions

Arithmetic expressions in Imp can be of three kinds: integer (\mathbb{Z}) constants, identifiers and sums. As anticipated, the evaluation of arithmetic expressions can fail, that is, the evaluation of arithmetic expression, conceptually, is not a total function. Again, the possible erroneous states we can get into when evaluating an arithmetic expression mainly concerns the use of undeclared identifiers and, as we did for stores, we can target the Maybe monad.

Without introducing them, we will use notations similar to that used earlier for commands ($\cdot \Downarrow \cdot$ and $\cdot \not\Downarrow$)

$\frac{}{\text{const } n \Downarrow \text{just } n}$	$\frac{\text{id} \in \sigma}{\text{var id} \Downarrow \sigma \text{id}}$	$\frac{a_1 \Downarrow \text{just } n_1 \quad a_2 \Downarrow \text{just } n_2}{\text{plus } a_1 a_2 \Downarrow \text{just } (n_1 + n_2)}$
$\frac{\text{id} \notin \sigma}{\text{var id} \not\Downarrow}$	$\frac{a_1 \not\Downarrow}{\text{plus } a_1 a_2 \not\Downarrow}$	$\frac{a_1 \Downarrow \text{just } n_1 \quad a_2 \not\Downarrow}{\text{plus } a_1 a_2 \not\Downarrow}$

Table 4: Inference rules for the semantics of arithmetic expressions of Imp

In Agda, these rules are implemented as shown in Listing 6.

```
aeval : ∀ (a : AExp) (s : Store) → Maybe ℤ
aeval (const x) s = just x
aeval (var x) s = s x
aeval (plus a a₁) s = aeval a s >>= λ v₁ → aeval a₁ s >>= λ v₂ → just (v₁ + v₂)
```

Listing 6: Agda interpreter for arithmetic expressions

5.2.2 Boolean expressions

Boolean expressions in Imp can be of four kinds: boolean constants, negation of a boolean expression, logical \wedge and, finally, comparison of arithmetic expressions. The line of reasoning for the definition of semantic rules follows what we underlined earlier, that is, we again target the Maybe monad.

$\frac{}{\text{const } c \Downarrow \text{just } c}$	$\frac{b \Downarrow c}{\neg b \Downarrow \neg c}$	$\frac{a_1 \Downarrow \text{just } n_1 \quad a_2 \Downarrow \text{just } n_2}{\text{le } a_1 a_2 \Downarrow \text{just } (n_1 < n_2)}$
$\frac{b_1 \Downarrow \text{just } c_1 \quad b_2 \Downarrow \text{just } c_2}{\text{and } b_1 b_2 \Downarrow \text{just } (c_1 \wedge c_2)}$	$\frac{b \Downarrow}{\neg b \Downarrow}$	$\frac{a_1 \Downarrow}{\text{le } a_1 a_2 \Downarrow}$
$\frac{a_1 \Downarrow \text{just } n_1 \quad a_2 \Downarrow}{\text{le } a_1 a_2 \Downarrow}$	$\frac{b_1 \Downarrow}{\text{and } b_1 b_2 \Downarrow}$	$\frac{b_1 \Downarrow \text{just } c_1 \quad b_2 \Downarrow}{\text{and } b_1 b_2 \Downarrow}$

Table 5: Inference rules for the semantics of boolean expressions of Imp

In Agda, these rules are implemented as shown in Listing 7.

```
beval : ∀ (b : BExp) (s : Store) → Maybe Bool
beval (const c) s = just c
beval (le a1 a2) s = aeval a1 s >>= λ v1 → aeval a2 s >>= λ v2 → just (v1 ≤b v2)
beval (not b) s = beval b s >>= λ b → just (bnot b)
beval (and b1 b2) s = beval b1 s >>= λ b1 → beval b2 s >>= λ b2 → just (b1 ∧ b2)
```

Listing 7: Agda interpreter for boolean expressions

5.2.3 Commands

The inference rules we give for commands follow the formalism of **big-step** operational semantics, that is, intermediate states of evaluation aren't shown explicitly in the rules themselves.

In Agda, these rules are implemented as shown in Listing 8.

5.2.4 Properties of the interpreter

Regarding the interpreter, the most important property we want to show puts in relation the starting store a command is evaluated in and the (hypothetical) resulting store. Up until now, we kept the mathematical layer and the code layer separated; from now on we will collapse the two and allow ourselves to use mathematical notation to express formal statements about the code: in practice, this means that, for example, the mathematical names *aeval*, *beval* and *ceval* refer to names from the code layer *aeval*, *beval* and *ceval*, respectively.

```

mutual
  ceval-while : ∀ {i} (c : Command) (b : BExp) (s : Store) → Thunk (Delay (Maybe Store)) i
  i
  ceval-while c b s = λ where .force → (ceval (while b c) s)

  ceval : ∀ {i} → (c : Command) → (s : Store) → Delay (Maybe Store) i
  ceval skip s = now (just s)
  ceval (assign id a) s = now (aeval a s >=>m λ v → just (update id v s))
  ceval (seq c c1) s = ceval c s >=>p λ s' → ceval c1 s'
  ceval (ifelse b c c1) s = now (beval b s) >=>p
    (λ bv → (if bv then ceval c s else ceval c1 s))
  ceval (while b c) s = now (beval b s) >=>p
    (λ bv →
      if bv then (ceval c s >=>p λ s' → later (ceval-while c b s))
      else now (just s))

```

Listing 8: Agda interpreter for commands

Theorem 5.2.4.1 (ceval does not remove identifiers): Let c be a command and σ_1 and σ_2 be two stores. Then

$$\text{ceval } c \sigma_1 \Downarrow \sigma_2 \rightarrow \sigma_1 \stackrel{u}{\sqsubseteq} \sigma_2 \quad (4)$$

In Agda:

```
cevalDownsqsubsetequ : ∀ (c : Command) (s s' : Store) (hDown : (ceval c s) ↓ s') → s ⊆u s'
```

Theorem 5.2.4.1 will be fundamental for later proofs.

5.3 Analyses and optimizations

We chose to demonstrate the use of coinduction in the definition of operational semantics implementing operations on the code itself (that is, they're static analyses), then showing proofs regarding the result of the execution of the program. The main inspiration for these operations is (Nipkow and Klein 2014).

5.3.1 Definite initialization analysis

The first operationnn we describe is **definite initialization analysis**. In general, the objective of this analysis is to ensure that no variable is ever used before being initialized, which is the kind of failure, among many, we chose to model.

Variables and indicator functions

This analysis deals with variables. Before delving into its details, we show first a function to compute the set of variables used in arithmetic and boolean expressions. The objective is to come up with a *set* of identifiers that appear in the expression: we chose to

represent sets in Agda using indicator functions, which we trivially define as parametric functions from a parametric set to the set of booleans, that is `CharacteristicFunction = A → Bool`; later, we will instantiate this type for identifiers, giving the resulting type the name of `VarsSet`. Foremost, we give a (parametric) notion of members equivalence (that is, a function `_==_ : A → A → Bool`); then, we equip indicator functions of the usual operations on sets: insertion, union, and intersection and define the usual property of inclusion.

```

 $\phi$  : CharacteristicFunction
 $\phi$  =  $\lambda$  _ → false

 $\_ \mapsto \_$  : (v : A) → (s : CharacteristicFunction) → CharacteristicFunction
(v  $\mapsto$  s) x = (v == x)  $\vee$  (s x)

 $\_ \cup \_$  : (s1 s2 : CharacteristicFunction) → CharacteristicFunction
(s1  $\cup$  s2) x = (s1 x)  $\vee$  (s2 x)

 $\_ \cap \_$  : (s1 s2 : CharacteristicFunction) → CharacteristicFunction
(s1  $\cap$  s2) x = (s1 x)  $\wedge$  (s2 x)

 $\_ \subseteq \_$  : (s1 s2 : CharacteristicFunction) → Set a
s1  $\subseteq$  s2 =  $\forall$  x → (x-in-s1 : s1 x  $\equiv$  true) → s2 x  $\equiv$  true

```

Listing 9: Implementation of indicator functions in Agda

Important properties of `CharacteristicFunctions` (and thus of `VarsSets`) follows.

Theorem 5.3.1.1.1 (Equivalence of indicator functions):

(using the **Axiom of extensionality**)

```

if-ext :  $\forall$  {s1 s2 : CharacteristicFunction} → (a-ex :  $\forall$  x → s1 x  $\equiv$  s2 x) → s1
 $\equiv$  s2

```

Theorem 5.3.1.1.2 (Neutral element of union):

```

 $\cup$ - $\phi$  :  $\forall$  {s : CharacteristicFunction} → (s  $\cup$   $\phi$ )  $\equiv$  s

```

Theorem 5.3.1.1.3 (Update inclusion):

```

 $\mapsto \subseteq$  :  $\forall$  {id} {s : CharacteristicFunction} → s  $\subseteq$  (id  $\mapsto$  s)

```

Theorem 5.3.1.1.4 (Transitivity of inclusion):

```

 $\subseteq\text{-trans} : \forall \{s_1 \ s_2 \ s_3 : \text{CharacteristicFunction}\} \rightarrow (s_1 \subseteq s_2 : s_1 \subseteq s_2) \\
\rightarrow (s_2 \subseteq s_3 : s_2 \subseteq s_3) \rightarrow s_1 \subseteq s_3$ 

```

We will also need a way to get a `VarsSet` from a `Store`, which is shown in Listing 10.

```

dom : Store → VarsSet
dom s x with (s x)
... | just _ = true
... | nothing = false

```

Listing 10: Code to compute the domain of a `Store` in Agda

Realization

Following (Nipkow and Klein 2014), the first formal tool we need is a means to compute the set of variables mentioned in expressions, shown in Listing 6; we also need a function to compute the set of variables that are definitely initialized in commands, which is shown in Listing 11.

```

avars : (a : AExp) → VarsSet
avars (const n) =  $\emptyset$ 
avars (var id) = id  $\mapsto$   $\emptyset$ 
avars (plus a1 a2) = (avars a1)  $\cup$  (avars a2)
bvars : (b : BExp) → VarsSet
bvars (const b) =  $\emptyset$ 
bvars (le a1 a2) = (avars a1)  $\cup$  (avars a2)
bvars (not b) = bvars b
bvars (and b b1) = (bvars b)  $\cup$  (bvars b1)

```

Listing 6: Agda code to compute variables in arithmetic and boolean expressions

```

cvars : (c : Command) → VarsSet
cvars skip =  $\emptyset$ 
cvars (assign id a) = id  $\mapsto$   $\emptyset$ 
cvars (seq c c1) = (cvars c)  $\cup$  (cvars c1)
cvars (ifelse b c c1) = (cvars c)  $\cap$  (cvars c1)
cvars (while b c) =  $\emptyset$ 

```

Listing 11: Agda code to compute initialized variables in commands

Theorem 5.2.4.1 allows us to show the following theorem.

Theorem 5.3.1.2.1 (ceval adds at least the variables in commands):

Let c be a command and σ_1 and σ_2 be two stores. Then

$$\text{ceval } c \ \sigma_1 \Downarrow \sigma_2 \rightarrow (\text{dom } \sigma_1 \cup (\text{cvars } c)) \subseteq (\text{dom } \sigma_2) \quad (5)$$

In Agda:

$$\begin{aligned} \text{ceval} \Downarrow \Rightarrow \text{sc} \subseteq s' : & \forall (c : \text{Command}) (s \ s' : \text{Store}) (h \Downarrow : (\text{ceval } c \ s) \Downarrow s') \\ & \rightarrow (\text{dom } s \cup (\text{cvars } c)) \subseteq (\text{dom } s') \end{aligned}$$

We now give inference rules that inductively build the relation that embodies the logic of the definite initialization analysis, shown in Table 7. In Agda, we define a datatype representing the relation of type $\text{Dia} : \text{VarsSet} \rightarrow \text{Command} \rightarrow \text{VarsSet} \rightarrow \text{Set}$, which is shown in Listing 12.

$$\begin{array}{c} \frac{}{\text{Dia } v \text{ skip } v} \qquad \frac{\text{avars } a \subseteq v}{\text{Dia } v (\text{assign id } a) (\text{id} \mapsto v)} \\ \frac{\text{Dia } v_1 \ c_1 \ v_2 \quad \text{Dia } v_2 \ c_2 \ v_3}{\text{Dia } v_1 (\text{seq } c_1 \ c_2) \ v_3} \quad \frac{\text{bvvars } b \subseteq v \quad \text{Dia } v \ c^t \ v^t \quad \text{Dia } v \ c^f \ v^f}{\text{Dia } v (\text{if } b \text{ then } c^t \text{ else } c^f) (v^t \cap v^f)} \\ \frac{\text{bvvars } b \subseteq v \quad \text{Dia } v \ c \ v_1}{\text{Dia } v (\text{while } b \ c) \ v} \end{array}$$

Table 7: Inference rules for the definite initialization analysis

```
data Dia : VarsSet → Command → VarsSet → Set where
  skip : ∀ (v : VarsSet) → Dia v (skip) v
  assign : ∀ a v id (a ⊆ v : (avars a) ⊆ v) → Dia v (assign id a) (id ↦ v)
  seq : ∀ v₁ v₂ v₃ c₁ c₂ → (relc₁ : Dia v₁ c₁ v₂) →
    (relc₂ : Dia v₂ c₂ v₃) → Dia v₁ (seq c₁ c₂) v₃
  if : ∀ b v vᵗ vᶠ cᵗ cᶠ (b ⊆ v : (bvvars b) ⊆ v) → (relcᶠ : Dia v cᶠ vᶠ) →
    (relcᵗ : Dia v cᵗ vᵗ) → Dia v (ifelse b cᵗ cᶠ) (vᵗ ∩ vᶠ)
  while : ∀ b v v₁ c → (b ⊆ v : (bvvars b) ⊆ v) →
    (relc : Dia v c v₁) → Dia v (while b c) v
```

Listing 12: Dia relation in Agda

What we want to show now is that if **Dia** holds, then the evaluation of a command c does not result in an error: while Theorem 5.3.1.2.2 and Theorem 5.3.1.2.3 show that if the variables in an arithmetic expression or a boolean expression are contained in a store the result of their evaluation can't be a failure (i.e. they result in “just” something), Theorem 5.3.1.2.4 shows that if **Dia** holds, then the evaluation of a program failing is absurd.

Theorem 5.3.1.2.2 (Soundness of definite initialization for arithmetic expressions):

$$\begin{aligned} \text{adia-sound} : & \forall (a : \text{AExp}) (s : \text{Store}) (\text{dia} : \text{avars } a \subseteq \text{dom } s) \\ & \rightarrow (\exists \lambda v \rightarrow \text{aeval } a \ s \equiv \text{just } v) \end{aligned}$$

Theorem 5.3.1.2.3 (Soundness of definite initialization for boolean expressions):

$\text{bdia-sound} : \forall (b : \text{BExp}) (s : \text{Store}) (\text{dia} : \text{bvars } b \subseteq \text{dom } s)$
 $\rightarrow (\exists \lambda v \rightarrow \text{beval } b \ s \equiv \text{just } v)$

Theorem 5.3.1.2.4 (Soundness of definite initialization for commands):

$\text{dia-sound} : \forall (c : \text{Command}) (s : \text{Store}) (v \ v' : \text{VarsSet}) (\text{dia} : \text{Dia } v \ c \ v')$
 $(v \subseteq s : v \subseteq \text{dom } s) \rightarrow (\text{h-err} : (\text{ceval } c \ s) \Downarrow) \rightarrow \perp$

Here, we show the proof of Theorem 5.3.1.2.4:

Proof:

```

dia-sound (assign id a) s v .(id ↦ v) (assign .a .v .id a⊆v) v⊆s h-err
  with (adia-sound a s (⊆-trans a⊆v v⊆s))
... | a' , eq-aeval rewrite eq-aeval rewrite eq-aeval with (h-err)
... | ()
dia-sound (ifelse b ct cf) s v .(vt n vf) (if .b .v vt vf .ct .cf b⊆v diaf diat)
v⊆s h-err
  with (bdia-sound b s λ x x-in-s1 → v⊆s x (b⊆v x x-in-s1))
... | false , eq-beval rewrite eq-beval rewrite eq-beval = dia-sound cf s v vf
diaf v⊆s h-err
dia-sound (ifelse b ct cf) s v .(vt n vf) (if .b .v vt vf .ct .cf b⊆v diaf diat)
v⊆s h-err
  | true , eq-beval rewrite eq-beval rewrite eq-beval = dia-sound ct s v vt diat
v⊆s h-err
dia-sound (seq c1 c2) s v1 v3 dia v⊆s h-err with dia
... | seq .v1 v2 .v3 .c1 .c2 dia-c1 dia-c2 with (ceval c1 s) in eq-ceval-c1
... | now nothing = dia-sound c1 s v1 v2 dia-c1 v⊆s (≡⇒≈ eq-ceval-c1)
... | now (just s') rewrite eq-ceval-c1 =
  dia-sound c2 s' v2 v3 dia-c2 (dia-ceval⇒⊆ dia-c1 v⊆s (≡⇒≈ eq-ceval-c1)) h-
err
dia-sound (seq c1 c2) s v1 v3 dia v⊆s h-err | seq .v1 v2 .v3 .c1 .c2 dia-c1 dia-
c2 | later x
  with (dia-sound c1 s v1 v2 dia-c1 v⊆s)
... | c1⊥ rewrite eq-ceval-c1 = dia-sound-seq-later c1⊥ dia-c2 h h-err
  where
    h : ∀ {s'} (h : (later x) ⊥ s') → v2 ⊆ dom s'
    h h1 rewrite (sym eq-ceval-c1) = dia-ceval⇒⊆ dia-c1 v⊆s h1
dia-sound (while b c) s v v' dia v⊆s h-err with dia
... | while .b .v v1 .c b⊆s dia-c
  with (bdia-sound b s (λ x x-in-s1 → v⊆s x (b⊆s x x-in-s1)))
... | false , eq-beval rewrite eq-beval = case h-err of λ ()

```



```

... | true , eq-beval with (ceval c s) in eq-ceval-c
... | now nothing = dia-sound c s v v1 dia-c vcs ( $\Rightarrow \approx$  eq-ceval-c)
dia-sound (while b c) s v v' dia vcs h-err | while .b .v v1 .c bcs dia-c
| true , eq-beval | now (just s') rewrite eq-beval rewrite eq-ceval-c
with h-err
... | later1 wz =
dia-sound (while b c) s' v v dia (ccs-trans vcs (ceval $\Downarrow$ cs c s s' ( $\Rightarrow \approx$  eq-ceval-
c))) wz
dia-sound (while b c) s v v' dia vcs h-err | while .b .v v1 .c bcs dia-c
| true , eq-beval | later x with (dia-sound c s v v1 dia-c vcs)
... | cz1 rewrite eq-beval rewrite eq-ceval-c = dia-sound-while-later cz1 dia
h h-err
where
h :  $\forall \{s'\} (h : (\text{later } x) \Downarrow s') \rightarrow v \subseteq \text{dom } s'$ 
h  $\{s'\}$  h1 rewrite (sym eq-ceval-c) = (ccs-trans vcs (ceval $\Downarrow$ cs c s s' h1))

```

□

5.3.2 Pure constant folding optimization

Pure constant folding is the second and last operation we considered. Again from (Nipkow and Klein 2014), the operation of pure folding consists in statically examining the source code of the program in order to move, when possible, computations from run-time to (pre-)compilation.

The objective of pure constant folding is that of finding all the places in the source code where the result of expressions is computable statically: examples of this situation are `true true`, `plus 1 1`, `le 0 1` and so on. This optimization is called *pure* because we avoid the passage of constant propagation, that is, we don't replace the value of identifiers even when their value is known at compile time.

Pure folding of arithmetic expressions

Pure folding optimization on arithmetic expressions is straightforward, and we define it as a function `apfold`. In words, what this optimization does is the following: let a be an arithmetic expression. Then, if a is a constant or an identifier the result of the optimization is a . If a is the sum of two other arithmetic expressions a_1 and a_2 ($a \equiv \text{plus } a_1 a_2$), the optimization is performed on the two immediate terms a_1 and a_2 , resulting in two potentially different expressions a'_1 and a'_2 . If both are constants v_1 and v_2 the result of the optimization is the constant $v_1 + v_2$; otherwise, the result of the optimization consists in the same arithmetic expression `plus a_1 a_2` left untouched. The Agda code for the function `apfold` is shown in Listing 13.

```

apfold : (a : AExp) → AExp
apfold (const x) = const x
apfold (var id) = var id
apfold (plus a1 a2) with (apfold a1) | (apfold a2)
... | const v1 | const v2 = const (v1 + v2)
... | a1' | a2' = plus a1' a2'

```

Listing 13: Agda code for pure folding of arithmetic expressions

Of course, what we want to show is that this optimization does not change the result of the evaluation (Theorem 5.3.2.1.1).

Theorem 5.3.2.1.1 (Soundness of pure folding for arithmetic expressions): Let a be an arithmetic expression and s be a store. Then

$$\text{aeval } a \ s \equiv \text{aeval } (\text{apfold } a) \ s \quad (6)$$

In Agda: `apfold-sound : ∀ a s → (aeval a s ≡ aeval (apfold a) s)`

Pure folding of boolean expressions

Pure folding of boolean expressions, which we define as a function `bpfold`, follows the same line of reasoning exposed in Chapter 5.3.2.1: let b be a boolean expression. If b is an expression with no immediates (i.e. $b \equiv \text{const } n$) we leave it untouched. If, instead, b has immediate subterms, we compute the pure folding of them and build a result accordingly, as shown in Listing 14.

```

bpfold : (b : BExp) → BExp
bpfold (const b) = const b
bpfold (le a1 a2) with (apfold a1) | (apfold a2)
... | const n1 | const n2 = const (n1 ≤b n2)
... | a1 | a2 = le a1 a2
bpfold (not b) with (bpfold b)
... | const n = const (lnot n)
... | b = not b
bpfold (and b1 b2) with (bpfold b1) | (bpfold b2)
... | const n1 | const n2 = const (n1 ∧ n2)
... | b1 | b2 = and b1 b2

```

Listing 14: Agda code for pure folding of arithmetic expressions

As before, our objective is to show that evaluating a boolean expressions after the optimization yields the same result as the evaluation without optimization.

Theorem 5.3.2.2.1 (Soundness of pure folding for boolean expressions): Let b be a boolean expression and s be a store. Then

$$\text{beval } b \ s \equiv \text{beval } (\text{bpfold } b) \ s \quad (7)$$

In Agda:

```
bpfold-sound : ∀ b s → (beval b s ≡ beval (bpfold b) s)
```

Pure folding of commands

Pure folding of commands builds on the definition of `apfold` and `bpfold` above combining the definitions as shown in Listing 15.

```
cpfold : Command → Command
cpfold skip = skip
cpfold (assign id a)
  with (apfold a)
... | const n = assign id (const n)
... | _ = assign id a
cpfold (seq c1 c2) = seq (cpfold c1) (cpfold c2)
cpfold (ifelse b c1 c2)
  with (bpfold b)
... | const false = cpfold c2
... | const true = cpfold c1
... | _ = ifelse b (cpfold c1) (cpfold c2)
cpfold (while b c) = while (bpfold b) (cpfold c)
```

Listing 15: Agda code for pure folding of commands

And, again, what we want to show is that the pure folding optimization does not change the semantics of the program, that is, optimized and unoptimized values converge to the same value or both diverge (Theorem 5.3.2.3.1).

Theorem 5.3.2.3.1 (Soundness of pure folding for commands): Let c be a command and s be a store. Then

$$\text{ceval } c \ s \equiv \text{ceval } (\text{cpfold } c) \ s \quad (8)$$

In Agda:

```
cpfold-sound : ∀ (c : Command) (s : Store)
→ ∞ ⊢ (ceval c s) ≈ (ceval (cpfold c) s)
```

Of course, what makes Theorem 5.3.2.3.1 different from the other soundness proofs in this chapter is that we cannot use propositional equality and we must instead use weak bisimilarity; we use the weak version as in terms of chains of `later` and `now`, if the

optimization does indeed change the syntactic tree of the command, if the evaluation converges to a value it may do so in a different number of steps; for example, the program `while 1 < 0 do skip` will be optimized to `while false do skip`, resulting in a shorter evaluation, as `1 < 0` will not be evaluated at runtime.

In this appendix we will show the Agda code for all the theorems mentioned in the thesis.

A.1 The partiality monad

A.1.1 Bisimilarity

Theorem 4.3.1

Proof:

```

reflexive : Reflexive R → ∀ {i} → Reflexive (Bisim R i)
reflexive refl^R {i} {now r}    = now refl^R
reflexive refl^R {i} {later rs} = later λ where .force → reflexive refl^R

symmetric : Sym P Q → ∀ {i} → Sym (Bisim P i) (Bisim Q i)
symmetric sym^PQ (now p)      = now (sym^PQ p)
symmetric sym^PQ (later ps) = later λ where .force → symmetric sym^PQ (ps .force)

transitive : Trans P Q R → ∀ {i} → Trans (Bisim P i) (Bisim Q i) (Bisim R i)
transitive trans^PQR (now p) (now q)    = now (trans^PQR p q)
transitive trans^PQR (later ps) (later qs) =
  later λ where .force → transitive trans^PQR (ps .force) (qs .force)

```

□

Theorem 4.3.2

Proof:

```

reflexive : Reflexive R → ∀ {i} → Reflexive (WeakBisim R i)
reflexive refl^R {i} {now x} = now refl^R
reflexive refl^R {i} {later x} = later λ where .force → reflexive (refl^R)

symmetric : Sym P Q → ∀ {i} → Sym (WeakBisim P i) (WeakBisim Q i)
symmetric sym^PQ (now x) = now (sym^PQ x)
symmetric sym^PQ (later x) = later λ where .force → symmetric (sym^PQ) (force x)
symmetric sym^PQ {i} (later1 x) = laterr (symmetric sym^PQ x)
symmetric sym^PQ (laterr x) = later1 (symmetric sym^PQ x)

```

□

Theorem 4.3.3

Proof:

```

left-identity : ∀ {i} (x : A) (f : A → Delay B i) → (now x) >=> f ≡ f x
left-identity {i} x f = _≡_.refl

right-identity : ∀ {i} (x : Delay A ∞) → i ⊢ x >=> now ≈ x
right-identity (now x) = now _≡_.refl
right-identity {i} (later x) = later (λ where .force → right-identity (force
x))

associativity : ∀ {i} {x : Delay A ∞} {f : A → Delay B ∞} {g : B → Delay C ∞}
→ i ⊢ (x >=> f) >=> g ≈ x >=> λ y → (f y >=> g)
associativity {i} {now x} {f} {g} with (f x)
... | now x1 = Codata.Sized.Delay.Bisimilarity.refl
... | later x1 = Codata.Sized.Delay.Bisimilarity.refl
associativity {i} {later x} {f} {g} = later (λ where .force → associativity {x
= force x})

```

□

A.2 The Imp programming language

A.2.1 Properties of stores

Theorem 5.1.2.1

Proof:

```

⊢u-trans : ∀ {s1 s2 s3 : Store} (h1 : s1 ⊢u s2) (h2 : s2 ⊢u s3) → s1 ⊢u s3
⊢u-trans h1 h2 id∈σ = h2 (h1 id∈σ)

```

□

A.2.2 Semantics

Theorem 5.2.4.1

Proof:

```

mutual
private
while-⊢u-later : ∀ {x : Thunk (Delay (Maybe Store)) ∞} (c : Command) (b : BExp)
(s s' : Store)
(f : ∀ (si : Store) → later x ↓ si → s ⊢u si) (h↓ : ((later x) >=>p (λ s →
later (ceval-while c b s)))) ↓ s')
→ s ⊢u s'
while-⊢u-later {x} c b s s' f (later1 h↓) {id} (z , id∈s)
with (force x) in eq-fx

```

```

... | now (just si) rewrite eq-fx
with h↓
... | later1 w↓
with (beval b si) in eq-b
... | just false with w↓
... | nowj refl = f si (later1 (⇒≈ eq-fx)) (z , id∈s)
while- $\sqsubseteq^u$ -later {x} c b s s' f (later1 h↓) {id} (z , id∈s) | now (just si) |
later1 w↓
| just true rewrite eq-b
with (bindxf↓⇒x↓ {x = ceval c si} {f = λ s → later (ceval-while c b s)}) w↓
... | si' , c↓si'
with (f si (later1 (⇒≈ eq-fx)) {id})
... | s $\sqsubseteq$ si
with (while- $\sqsubseteq^u$  c b si s' (λ { s1 s1i c↓si {id} (z' , id∈s1) → ceval↓⇒ $\sqsubseteq^u$  c s1
s1i c↓si {id} (z' , id∈s1)}) w↓ {id})
... | si $\sqsubseteq$ s' =  $\sqsubseteq^u$ -trans s $\sqsubseteq$ si si $\sqsubseteq$ s' {id} (z , id∈s)
while- $\sqsubseteq^u$ -later {x} c b s s' f (later1 h↓) {id} (z , id∈s)
| later x1 = while- $\sqsubseteq^u$ -later {x1} c b s s' (λ { si x2 x3 → f si (later1 (⇒≈↓
eq-fx x2)) x3}) h↓ {id} (z , id∈s)

```

```

while- $\sqsubseteq^u$  : ∀ (c : Command) (b : BExp) (s s' : Store) (f : ∀ (s si : Store) →
(ceval c s) ↓ si → s  $\sqsubseteq^u$  si)
(h↓ : ((ceval c s) ⇒p (λ s → later (ceval-while c b s)))) ↓ s') → s  $\sqsubseteq^u$  s'
while- $\sqsubseteq^u$  c b s s' f h↓ {id}
with (ceval c s) in eq-c
... | now (just si)
with (f s si (⇒≈ eq-c))
... | s $\sqsubseteq$ si rewrite eq-c
with h↓
... | later1 w↓
with (beval b si) in eq-b
... | just false rewrite eq-b with w↓
... | nowj refl = s $\sqsubseteq$ si {id}
while- $\sqsubseteq^u$  c b s s' f h↓ {id} | now (just si) | s $\sqsubseteq$ si | later1 w↓ | just true
rewrite eq-b
=  $\sqsubseteq^u$ -trans {s} {si} {s'} s $\sqsubseteq$ si (while- $\sqsubseteq^u$  c b si s' f w↓) {id}
while- $\sqsubseteq^u$  c b s s' f h↓
| later x
with h↓
... | later1 w↓ = while- $\sqsubseteq^u$ -later {x} c b s s' (λ { si x1 x2 → f s si (⇒≈↓ eq-
c x1) x2}) h↓

```

```

ceval↓⇒ $\sqsubseteq^u$  : ∀ (c : Command) (s s' : Store) (h↓ : (ceval c s) ↓ s') → s  $\sqsubseteq^u$  s'

```

```

ceval $\Downarrow \Rightarrow \sqsubseteq^u$  skip s .s (nowj refl) x = x
ceval $\Downarrow \Rightarrow \sqsubseteq^u$  (assign id a) s s' h $\Downarrow$  {id1} x
  with (aeval a s)
... | just v
  with h $\Downarrow$ 
... | nowj refl
  with (id = id1) in eq-id
... | true rewrite eq-id = v , refl
... | false rewrite eq-id = x
ceval $\Downarrow \Rightarrow \sqsubseteq^u$  (ifelse b ct cf) s s' h $\Downarrow$  x
  with (beval b s) in eq-b
... | just true rewrite eq-b = ceval $\Downarrow \Rightarrow \sqsubseteq^u$  ct s s' h $\Downarrow$  x
... | just false rewrite eq-b = ceval $\Downarrow \Rightarrow \sqsubseteq^u$  cf s s' h $\Downarrow$  x
ceval $\Downarrow \Rightarrow \sqsubseteq^u$  (seq c1 c2) s s' h $\Downarrow$  {id}
  with (bindxf $\Downarrow \Rightarrow x \Downarrow$  {x = ceval c1 s} {f = ceval c2} h $\Downarrow$ )
... | si , c1 $\Downarrow$ si
  with (bindxf $\Downarrow - x \Downarrow \Rightarrow f \Downarrow$  {x = ceval c1 s} {f = ceval c2} h $\Downarrow$  c1 $\Downarrow$ si)
... | c2 $\Downarrow$ s' =  $\sqsubseteq^u$ -trans (ceval $\Downarrow \Rightarrow \sqsubseteq^u$  c1 s si c1 $\Downarrow$ si {id}) (ceval $\Downarrow \Rightarrow \sqsubseteq^u$  c2 si s'
c2 $\Downarrow$ s' {id}) {id}
ceval $\Downarrow \Rightarrow \sqsubseteq^u$  (while b c) s s' h $\Downarrow$  {id} x
  with (beval b s) in eq-b
... | just false with h $\Downarrow$ 
... | nowj refl = x
ceval $\Downarrow \Rightarrow \sqsubseteq^u$  (while b c) s s' h $\Downarrow$  {id} x
  | just true rewrite eq-b = while- $\sqsubseteq^u$  c b s s' (λ s1 s2 h → ceval $\Downarrow \Rightarrow \sqsubseteq^u$  c s1 s2
h) h $\Downarrow$  {id} x

```

□

Bibliography

- Abel, Andreas, and Chapman, James. “Normalization by Evaluation in the Delay Monad: A Case Study for Coinduction Via Copatterns and Sized Types.” *Electronic Proceedings in Theoretical Computer Science*, vol. 153, doi:10.4204/eptcs.153.4.
- Capretta, Venanzio. “General Recursion Via Coinductive Types.” *Logical Methods in Computer Science*, doi:10.2168/lmcs-1(2:1)2005.
- Chapman, James, et al. “Quotienting the Delay Monad by Weak Bisimilarity.” *Theoretical Aspects of Computing - ICTAC 2015*, Springer International Publishing, doi: 10.1007/978-3-319-25150-9_8.
- Danielsson, Nils Anders. “Operational Semantics Using the Partiality Monad.” *Proceedings of the 17th ACM SIGPLAN International Conference on Functional Programming*, ACM, doi:10.1145/2364527.2364546.
- Kohl, Christina, and Schwaiger, Christina. “Monads in Computer Science.” <https://ncatlab.org/nlab/files/KohlSchwaiger-Monads.pdf>.
- Moggi, E. “Computational Lambda-Calculus and Monads.” [1989] *Proceedings. Fourth Annual Symposium on Logic in Computer Science*, vol. 0, doi:10.1109/LICS.1989.39155.
- Nipkow, Tobias, and Klein, Gerwin. *Concrete Semantics: With Isabelle/hol*. Springer Publishing Company, 2014.
- Pierce, Benjamin C., et al. *Logical Foundations*. 2023.